Dror Bar-Natan: Talks: UofT-GS-090930:

Dror Bar-Natan: Academic Pensieve: 2009-09:

Hilbert's 13th Problem, the First Step

Pensieve Header: The first step of the Kolmogorov-Arnol'd solution of Hilbert's 13th problem.

```
SetOptions[Rasterize, RasterSize → 1600, ImageSize → 1600];
SetOptions[Plot3D, PlotPoints → 150, ColorFunction → Hue,
  ColorFunctionScaling → False, ViewPoint → {-1, -2, 1}];
SetOptions[DensityPlot, PlotPoints → 150, ColorFunction → Hue,
  ColorFunctionScaling → False];
```

```
f[x_, y_] := N[
   Re[Zeta[x + I (13 + 4 y)]] / 3
  ];
Rasterize[GraphicsRow[{
   Plot3D[f[x, y], {x, 0, 1}, {y, 0, 1}],
   DensityPlot[f[x, y], {x, 0, 1}, {y, 0, 1}]
  }]]
```



$$\left\{\text{lambda} = N\left[\frac{-1+\sqrt{5}}{2}\right],\ n = 5,\ \text{eps} = 0.1\right\}$$

```
{0.618034, 5, 0.1}
```

```
Lattice = Table[
  Point[(i - 0.5) / n, (j - 0.5) / n],
  {i, 1, n}, {j, 1, n}
 ]
```

```
{{Point[0.1, 0.1], Point[0.1, 0.3], Point[0.1, 0.5], Point[0.1, 0.7], Point[0.1, 0.9]},
 {Point[0.3, 0.1], Point[0.3, 0.3], Point[0.3, 0.5], Point[0.3, 0.7], Point[0.3, 0.9]},
 {Point[0.5, 0.1], Point[0.5, 0.3], Point[0.5, 0.5], Point[0.5, 0.7], Point[0.5, 0.9]},
 {Point[0.7, 0.1], Point[0.7, 0.3], Point[0.7, 0.5], Point[0.7, 0.7], Point[0.7, 0.9]},
 {Point[0.9, 0.1], Point[0.9, 0.3], Point[0.9, 0.5], Point[0.9, 0.7], Point[0.9, 0.9]}}
```

```
ProjectionPoints = Sort[Flatten[Lattice /. Point[x_, y_] :> x + lambda * y]]
```

```
{0.161803, 0.28541, 0.361803, 0.409017, 0.48541, 0.532624, 0.561803, 0.609017,
 0.656231, 0.68541, 0.732624, 0.761803, 0.809017, 0.856231, 0.88541, 0.932624,
 0.961803, 1.00902, 1.05623, 1.08541, 1.13262, 1.20902, 1.25623, 1.33262, 1.45623}
```

```
MinGap = Min[Abs[ProjectionPoints - RotateRight[ProjectionPoints]]]
```

```
0.0291796
```

```
Slope = 2 / 3 * MinGap * n / (1 + lambda)
```

```
0.0601133
```

```
Phi = PLFunction @@ Flatten[{
    Point[0, 0],
    Table[
     {
      Point[(i - 1 + eps) / n, (i - 1 / 2) / n - Slope * (1 / 2 - eps) / n] ,
      Point[(i - eps) / n, (i - 1 / 2) / n + Slope * (1 / 2 - eps) / n]
     },
     {i, n}
    ],
    Point[1, 1]
   }
  ]
```

```
PLFunction[Point[0, 0], Point[0.02, 0.0951909], Point[0.18, 0.104809], Point[0.22, 0.295191],
 Point[0.38, 0.304809], Point[0.42, 0.495191], Point[0.58, 0.504809], Point[0.62, 0.695191],
 Point[0.78, 0.704809], Point[0.82, 0.895191], Point[0.98, 0.904809], Point[1, 1]]
```

```
PLFunction[pts__Point][x_] := Module[
  {i = 1, x0, x1, xi0, xi1},
  While[{pts}[[i, 1]] < x, ++i];
  If[i == 1,
   {pts}[[1, 1]],
   (* else *) {x0, xi0} = List @@ {pts}[[i - 1]];
   {x1, xi1} = List @@ {pts}[[i]];
   xi0 + (x - x0) * (xi1 - xi0) / (x1 - x0)
  ]
 ]
```

```
Plot[Phi[x], {x, 0, 1}]
```




```
Rasterize[Plot3D[Phi[x] + lambda * Phi[y], {x, 0, 1}, {y, 0, 1}, ColorFunction → Automatic]]
```



```
g = PLFunction @@ Sort[Flatten[{
      Point[0, f[0, 0]],
      Table[
       {
        Point[
         ((i - 1 / 2) / n - Slope * (1 / 2 - eps) / n) + lambda ((j - 1 / 2) / n - Slope * (1 / 2 - eps) / n),
         f[(i - 1 / 2) / n, (j - 1 / 2) / n]
        ],
        Point[
         ((i - 1 / 2) / n + Slope * (1 / 2 - eps) / n) + lambda ((j - 1 / 2) / n + Slope * (1 / 2 - eps) / n),
         f[(i - 1 / 2) / n, (j - 1 / 2) / n]
        ]
       },
       {i, n}, {j, n}
      ],
      Point[1 + lambda, f[1, 1]]
     }]]
```

```
PLFunction[Point[0, 0.0880162], Point[0.154022, -0.0000900936],
 Point[0.169585, -0.0000900936], Point[0.277629, -0.127398],
 Point[0.293191, -0.127398], Point[0.354022, 0.0411255], Point[0.369585, 0.0411255],
 Point[0.401236, -0.0754642], Point[0.416798, -0.0754642], Point[0.477629, -0.0598832],
 Point[0.493191, -0.0598832], Point[0.524843, 0.190866], Point[0.540405, 0.190866],
 Point[0.554022, 0.0776739], Point[0.569585, 0.0776739], Point[0.601236, -0.00738564],
 Point[0.616798, -0.00738564], Point[0.648449, 0.587064], Point[0.664012, 0.587064],
 Point[0.677629, -0.00227207], Point[0.693191, -0.00227207], Point[0.724843, 0.222172],
 Point[0.740405, 0.222172], Point[0.754022, 0.109972], Point[0.769585, 0.109972],
 Point[0.801236, 0.0490366], Point[0.816798, 0.0490366], Point[0.848449, 0.553098],
 Point[0.864012, 0.553098], Point[0.877629, 0.0468615], Point[0.893191, 0.0468615],
 Point[0.924843, 0.246573], Point[0.940405, 0.246573], Point[0.954022, 0.138431],
 Point[0.969585, 0.138431], Point[1.00124, 0.0958302], Point[1.0168, 0.0958302],
 Point[1.04845, 0.523159], Point[1.06401, 0.523159], Point[1.07763, 0.0887491],
 Point[1.09319, 0.0887491], Point[1.12484, 0.265583], Point[1.14041, 0.265583],
 Point[1.20124, 0.13467], Point[1.2168, 0.13467], Point[1.24845, 0.496932],
 Point[1.26401, 0.496932], Point[1.32484, 0.280389], Point[1.34041, 0.280389],
 Point[1.44845, 0.474073], Point[1.46401, 0.474073], Point[1.61803, 0.541429]]
```

```
Plot[
 g[zeta], {zeta, 0, 1 + lambda},
 PlotPoints → 200, ColorFunction → (Hue[#2] &), ColorFunctionScaling → False
]
```



```
Rasterize[DensityPlot[
  g[zeta], {zeta, 0, 1 + lambda}, {dummy, 0, 0.1},
  PlotPoints → {400, 2}, AspectRatio → 1 / 10
 ]]
```




```
Rasterize[GraphicsRow[{
   DensityPlot[g[Phi[x] + lambda * Phi[y]], {x, 0, 1}, {y, 0, 1}],
   DensityPlot[f[x, y], {x, 0, 1}, {y, 0, 1}]
  }]]
```



```
DoubleStairs = Rasterize[Plot3D[
    Phi[x] + lambda * Phi[y], {x, 0, 1}, {y, 0, 1},
    ColorFunction → (Hue[g[#3]] &), PlotPoints → 300
  ]]
```




```
Export[
 "C:/drorbn/AcademicPensieve/2009-09/DoubleStairs.jpg",
 ImageCrop[DoubleStairs]
]
```

```
C:/drorbn/AcademicPensieve/2009-09/DoubleStairs.jpg
```

```
Export[
 "C:/drorbn/AcademicPensieve/2009-09/DoubleStairs_640.jpg",
 ImageResize[ImageCrop[DoubleStairs], 640]
]
```

```
C:/drorbn/AcademicPensieve/2009-09/DoubleStairs_640.jpg
```